
Introduzione 

envenuto in “250 formule spiegate passo-passo”.  

Excel è il leader del mercato dei fogli di calcolo, di gran lunga. Questo 

non è solo dovuto all'enorme potenziale di marketing di Microsoft, ma 

perché è veramente il miglior foglio di calcolo disponibile.  

Un'area in cui la superiorità di Excel è più evidente che mai, è quella delle formule. 

Come vedrai, Excel consente di fare cose con le formule che sono impossibili con 

altri fogli di calcolo.  

È una scommessa sicura che solo il 10 per cento degli utenti di Excel sa veramente 

come ottenere il massimo dalle formule. In questo libro cerchiamo di portarti in 

quel gruppo di élite. Sei pronto?  

Questo non è un libro per principianti di Excel. Se non hai assolutamente 

esperienza con Excel, probabilmente questo non è il miglior libro per te. 

Questo libro contiene 250 formule dettagliate e spiegate in ogni passaggio. 

Consideralo un prontuario in cui trovare già pronta la formula che ti serve. 

Le formule vanno da “semplici” a “notevolmente complesse” e queste ultime, 

senza una solida conoscenza di base, potrebbero risultare ostiche. Per questo, 

consigliamo la lettura preventiva del libro “Formule e funzioni di Microsoft 

Excel”, del quale il presente libro è un naturale complemento. 

B 

https://excelacademy.it/book/formule-funzioni-microsoft-excel-2/
https://excelacademy.it/book/formule-funzioni-microsoft-excel-2/


Capitolo 1 - Formule per contare 

 
13 

Capitolo 1 - Formule per contare 

 



  

 
Excel Academy 14 

1. Contare le date comprese tra due estremi 

 

 

Per contare il numero di celle che contengono delle date comprese tra due estremi, 

è possibile utilizzare la funzione CONTA.PIÙ.SE. Nell'esempio mostrato, la cella 

F3 contiene questa formula: 

=CONTA.PIÙ.SE(date; 

">="&DATA(E3;1;1);date;"<="&DATA(E3;12;31)) 

 

Questa formula conta le date di nascita del 2003 che appaiono nell'intervallo 

C3:C12. 

Nota: questa formula utilizza un intervallo denominato "date" che rappresenta 

C3:C12. 

 

La funzione CONTA.PIÙ.SE è progettata per contare le celle che soddisfano più 

criteri. In questo caso, dobbiamo fornire due criteri: un criterio per la data iniziale 

e uno per la data finale. 

Per costruire le due date, utilizziamo la funzione DATA: 

DATA(E3;1;1) // restituisce il primo giorno 

dell’anno 

DATA(E3;12;31) // restituisce l’ultimo giorno 

dell’anno 

 

La funzione DATA consente di creare facilmente le date in base agli argomenti 

anno, mese e giorno che possono essere inseriti o forniti come riferimenti di cella. 

Nell'esempio, il mese e il giorno sono inseriti, mentre prendiamo l'anno dalla 

colonna E. 

Si noti che “>=” e “<=” devono essere inseriti come testo e racchiusi da doppie 

virgolette. Ciò significa che dobbiamo utilizzare la concatenazione (&) per unire 

ogni operatore alla data. 

2. Contare i numeri compresi tra due estremi 

 

 

Per contare il numero di celle che contengono valori tra due numeri, è possibile 

utilizzare la funzione CONTA.PIÙ.SE. 

Nell'esempio, la cella attiva contiene questa formula: 

=CONTA.PIÙ.SE(C3:C12;">=75";C3:C12;"<=90") 

 

La funzione CONTA.PIÙ.SE è progettata per contare le celle che soddisfano più 

criteri. In questo caso, poiché forniamo lo stesso intervallo per due criteri, ogni 

cella dell’intervallo deve soddisfare entrambi i criteri per essere conteggiata. 

 

Con la funzione CONTA.SE 

Se hai una versione di Excel precedente al 2007, che non dispone della funzione 

CONTA.PIÙ.SE, è possibile utilizzare la funzione CONTA.SE in questo modo: 

=CONTA.SE(C3:C12;">=75")-CONTA.SE(C3:C12;">90") 

 

Il primo CONTA.SE conta il numero di celle con valore maggiore o uguale al 

limite inferiore (che per definizione include anche valori maggiori del limite 

superiore). Il secondo CONTA.SE conta il numero di celle con valori maggiori al 

limite superiore. Questo secondo numero viene quindi sottratto dal primo numero, 

producendo il risultato finale, ossia il numero di celle che contengono valori tra 75 

e 90. 

 

Rendere “variabile” il criterio 

Se desideri utilizzare come parte dei criteri un valore presente in un'altra cella, 

utilizza il carattere “e commerciale” (&) per concatenare, così: 

CONTA.SE(C3:C12; ">"&F3) 

 

Se il valore nella cella F3 fosse "75", dopo la concatenazione il criterio 

diventerebbe ">75". 

3. Contare le celle uguali a 

 

 

Per contare il numero di celle uguali a valori specifici, è possibile utilizzare la 

funzione CONTA.SE. 

Nell'esempio mostrato, G5 contiene questa formula: 

=CONTA.SE(D3:D12;"rosso") 

 

La funzione CONTA.SE è completamente automatica: conta il numero di celle di 

un intervallo che corrispondono al criterio fornito. 

Come intervallo usiamo D3:D12, che contiene i colori. Come criterio, utilizziamo 

semplicemente "rosso". 

CONTA.SE restituisce il conteggio dei valori in D3:D12 che sono uguali a 

"rosso". 

 

Nota: quando i valori di testo vengono forniti direttamente come criteri, devono 

essere racchiusi tra doppie virgolette (""). Se i criteri sono già disponibili in un'altra 

cella, è possibile fornire come criterio l'indirizzo della cella senza virgolette, come 

si vede in CONTEGGIO DI RIEPILOGO CON CONTA.SE. 

4. Conteggio di riepilogo con CONTA.SE 

 

 

Quando si lavora con i dati, un'esigenza comune è quella di eseguire calcoli di 

riepilogo che mostrino i conteggi totali in vari modi. Ad esempio, un conteggio 

totale per categoria, per colore, per dimensione, per stato, ecc. La funzione 

CONTA.SE è un buon modo per generare questi tipi di riepiloghi. 

 

Se hai un numero limitato di valori da contare questa è una buona soluzione. 

Tuttavia, se lavori con un vasto elenco di valori, che cambiano nel tempo, una 

tabella pivot è un'opzione migliore. Per sfruttare al massimo le potenzialità delle 

Tabelle Pivot, ti consigliamo la lettura del nostro libro “Tabelle Pivot – La guida 

definitiva”. 

 

Nell'esempio mostrato, abbiamo un piccolo insieme di dati. Tutti gli ordini 

riguardano magliette, prodotte in 5 diversi colori: verde, rosso, giallo, arancio e blu. 

Utilizziamo CONTA.SE per calcolare la ripartizione per colore. La formula nella 

cella G3 è: 

https://excelacademy.it/book/tabelle-pivot/
https://excelacademy.it/book/tabelle-pivot/

=CONTA.SE(colori;F3) 

 

dove "colori" è un intervallo denominato per le celle D3:D12. In questo caso 

utilizziamo un intervallo denominato per rendere la formula facile da copiare lungo 

la tabella riepilogativa. In alternativa, è possibile utilizzare un riferimento assoluto 

come in questo caso: 

=CONTA.SE($D$3:$D$12;F3) 

 

La funzione CONTA.SE richiede due argomenti: un intervallo di celle da contare e 

il criterio da utilizzare per il conteggio. In questo caso vogliamo contare per colore, 

quindi abbiamo impostato una piccola tabella che elenca tutti i colori in colonna F. 

Ciò consente di utilizzare i nomi dei colori di colonna F sia come etichette, sia 

come criterio che entra in CONTA.SE come secondo argomento. 

In questo caso stiamo testando l'uguaglianza di valori, quindi non abbiamo 

bisogno di utilizzare operatori logici. Possiamo semplicemente immettere il campo 

denominato "colori" come intervallo e un riferimento alla cella adiacente di 

colonna F come criterio. 

5. Contare celle uguali a un valore, tenendo 

conto delle differenze tra maiuscolo e 

minuscolo 

 

 

Per contare le celle che contengono un determinato testo tenendo conto delle 

differenze tra maiuscole e minuscole, è possibile utilizzare una formula che utilizza 

la funzione IDENTICO insieme a MATR.SOMMA.PRODOTTO. 

Nell'esempio, c'è un elenco di nomi in un intervallo denominato "nomi" (B3:B12). 

Nell'intervallo D3:D6, c'è un altro elenco di nomi, con duplicati rimossi. Per 

ottenere il conteggio di "Luigi", è possibile utilizzare la formula: 

=MATR.SOMMA.PRODOTTO(--IDENTICO(D3;B3:B12)) 

 

Nell'esempio, che utilizza l’intervallo denominato "nomi", la prima formula è: 

=MATR.SOMMA.PRODOTTO(--IDENTICO(D3;nomi)) 

Entrambe le formule riportate sopra restituiscono 1, in quanto c’è un’unica 

occorrenza di "Luigi" (con l’iniziale maiuscola) nell'intervallo B3:B12. 

La funzione IDENTICO richiede due argomenti, testo1 e testo2. Quando questi 

valori corrispondono esattamente (rispettando le maiuscole), IDENTICO 

restituisce VERO. In questo caso, stiamo dando a IDENTICO un intervallo per 

testo1, quindi la nostra diventa una formula matriciale e restituisce una matrice di 

valori VERO e FALSO come questa: 

{VERO.FALSO.FALSO.FALSO.FALSO.FALSO.FALSO.FALSO.FA

LSO.FALSO} 

 

Ogni VERO rappresenta una corrispondenza esatta di "Luigi" in B3:B12. 

Il doppio meno (tecnicamente chiamato doppio unario) converte i valori VERO e 

FALSO in 1 e 0 in modo che la matrice precedente diventa così: 

{1.0.0.0.0.0.0.0.0.0} 

 

Infine, MATR.SOMMA.PRODOTTO somma semplicemente i valori nella 

matrice e restituisce 1. Poiché MATR.SOMMA.PRODOTTO può gestire le 

matrici in modo nativo, non è necessario utilizzare Ctrl + Maiusc + Invio per 

immettere questa formula. 

6. Contare le celle uguali a x oppure a y 

 

 

Per contare il numero di celle uguali ad un valore oppure ad un altro, si utilizza una 

formula che utilizza due volte la funzione CONTA.SE. 

Nell'esempio, la cella attiva contiene questa formula: 

=CONTA.SE(B3:B12;"mele") + CONTA.SE(B3:B12;"pere") 

 

CONTA.SE conta il numero di celle che, in un intervallo, corrispondono al 

criterio fornito. In questo caso, il criterio per il primo CONTA.SE è "mele" e il 

criterio per il secondo CONTA.SE è "pere". Il primo CONTA.SE restituisce il 

conteggio delle celle che, in B4:B12, sono uguali a "mele". Il secondo CONTA.SE 

restituisce il numero delle celle che, in B4:B12 sono uguali a "pere". Questi due 

conteggi vengono sommati insieme e la somma viene restituita come risultato della 

formula. 

Si noti che i valori di testo nei criteri di CONTA.SE devono essere racchiusi tra 

virgolette (""). Tieni anche presente che CONTA.SE non è mai sensibile alla 

distinzione tra maiuscole e minuscole. Nell'esempio, le parole "mele" e "pere" in 

qualsiasi combinazione di lettere maiuscole e minuscole saranno conteggiate. 

Se vuoi contare celle che contengono valori numerici, non è necessario aggiungere 

virgolette intorno ai numeri. Ad esempio, se nell'esempio precedente avessi voluto 

contare delle celle che contenevano 0 o 1, la formula sarebbe stata: 

=CONTA.SE(B3:B12;0) + CONTA.SE(B3:B12;1) 

 

Se invece di contare le celle uguali a X oppure Y, volessi contare le celle che 

contengono X oppure Y vedi CONTARE LE CELLE CONTENENTI DEL TESTO 

SPECIFICO. 

7. Contare le celle contenenti del testo 

specifico 

 

 

Per contare il numero di celle che contengono un determinato testo, è possibile 

utilizzare la funzione CONTA.SE. 

Nell'esempio, la cella attiva contiene questa formula: 

=CONTA.SE(B3:B12;"*a*") 

 

CONTA.SE conta il numero di celle nell'intervallo che contengono "a", 

confrontando il contenuto di ciascuna cella con lo schema "*a*", fornito come 

criterio. Il simbolo "*" (asterisco) è un carattere jolly che in Excel significa 

"qualsiasi numero di caratteri", questo modello quindi intercetta ogni cella che 

contiene "a" in qualsiasi posizione. Il conteggio delle celle che corrispondono a 

questo schema viene restituito come numero. 

È possibile aggiustare facilmente questa formula per utilizzare come criterio il 

contenuto di un'altra cella. Ad esempio, se A1 contiene il testo che si desidera 

confrontare, utilizza la formula: 

=CONTA.SE(intervallo;"*"&a1&"*") 

Versione che distingue tra maiuscole e minuscole 

Se hai bisogno di una versione sensibile alla distinzione tra maiuscole e minuscole, 

non puoi utilizzare CONTA.SE. È invece possibile confrontare ogni cella 

dell'intervallo utilizzando una formula basata sulla funzione TROVA e sulla 

funzione VAL.NUMERO, come spiegato in VERIFICARE SE UNA CELLA 

CONTIENE DEL TESTO SPECIFICO. 

TROVA è sensibile alla distinzione tra maiuscole e minuscole, puoi poi utilizzare 

MATR.SOMMA.PRODOTTO per contare i risultati. La formula sarebbe simile a 

questa: 

=MATR.SOMMA.PRODOTTO(--

(VAL.NUMERO(TROVA(testo;intervallo)))) 

 

Dove testo è il testo che stai cercando e intervallo è l'intervallo di celle da contare. 

Non è necessario utilizzare i caratteri jolly, poiché TROVA restituirà un numero se 

testo viene trovato in qualsiasi punto della cella. 

8. Contare le celle uguali a una tra tante cose 

 

 

Per contare il numero di celle uguali a uno tra tanti valori, è possibile utilizzare la 

funzione CONTA.SE all'interno di MATR.SOMMA.PRODOTTO. 

Nell'esempio mostrato, la cella F3 contiene questa formula: 

=MATR.SOMMA.PRODOTTO(CONTA.SE(B3:B12; scelta)) 

 

Dove per scelta si intende un intervallo denominato che rappresenta le celle D3:D5. 

Nota: CONTA.SE non distingue tra maiuscole e minuscole 

 

CONTA.SE conta il numero di celle dell’intervallo che soddisfano il criterio 

fornito 

. Quando a CONTA.SE si assegna come criterio (secondo argomento) un 

intervallo, essa restituisce come risultato una serie di numeri, in cui ogni numero 

rappresenta il conteggio di un elemento dell’intervallo. In questo caso, l'intervallo 

denominato "scelta" (D3:D5) contiene 3 valori, quindi CONTA.SE restituisce 3 

risultati in una matrice come questa: 

{2.2.1} 

Per gestire questa matrice, utilizziamo la funzione MATR.SOMMA.PRODOTTO, 

progettata per lavorare con le matrici. MATR.SOMMA.PRODOTTO somma 

semplicemente gli elementi della matrice e restituisce il risultato, 5. 

Con una costante di matrice 

Alternativamente, con un numero limitato di valori, è possibile anche utilizzare una 

costante di matrice in una formula con la funzione SOMMA, in questo modo: 

=SOMMA(CONTA.SE 

(B3:B12;{"giallo"."verde"."nero"})) 

 

Ma se utilizzi i riferimenti di cella come criterio, dovrai inserirla come formula 

matriciale o passare a MATR.SOMMA.PRODOTTO. 

9. Contare le celle maggiori di 

 

 

Per contare il numero di celle che hanno valori maggiori di un determinato 

numero, è possibile utilizzare la funzione CONTA.SE. 

Nell'esempio mostrato, la cella attiva contiene questa formula: 

=CONTA.SE(C3:C12;">80") 

 

CONTA.SE conta il numero di celle dell'intervallo che contengono valori numerici 

maggiori di 80 e restituisce il risultato come numero. 

Se come criterio desideri utilizzare un valore presente in un'altra cella, utilizza il 

carattere “e commerciale” (&) per concatenare, in questo modo: 

=CONTA.SE(intervallo;">"&A1) 

 

Se il valore della cella A1 è "70", dopo la concatenazione il criterio sarà ">70". 

10. Contare le celle minori di 

 

 

Per contare il numero di celle che hanno valori minori di un determinato numero, 

è possibile utilizzare la funzione CONTA.SE. 

Nell'esempio mostrato, la cella attiva contiene questa formula: 

=CONTA.SE(C3:C12;"<80") 

 

CONTA.SE conta il numero di celle dell'intervallo che contengono valori numerici 

minori di 80 e restituisce il risultato come numero. 

Se come criterio desideri utilizzare un valore presente in un'altra cella, utilizza il 

carattere “e commerciale” (&) per concatenare, in questo modo: 

=CONTA.SE(intervallo;"<"&A1) 

 

Se il valore della cella A1 è "90", dopo la concatenazione il criterio sarà "<90". 

11. Contare le celle diverse da 

 

 

Per contare il numero di celle che contengono valori diversi da un determinato 

valore, è possibile utilizzare la funzione CONTA.SE. 

Nell'esempio, la cella attiva contiene questa formula: 

=CONTA.SE(D3:D12;"<>Completato") 

 

CONTA.SE conta il numero di celle dell’intervallo che soddisfano il criterio 

fornito. 

Nell'esempio utilizziamo "<>" (l'operatore logico per "diverso da") per contare le 

celle nell'intervallo D3:D12 che non sono uguali a "completato". CONTA.SE 

restituisce il conteggio come risultato. 

CONTA.SE non è sensibile alla distinzione tra maiuscole e minuscole. In questo 

esempio, la parola "completato" può essere visualizzata in qualsiasi combinazione 

di lettere maiuscole / minuscole e non verrà conteggiata. 

Se come criterio vuoi utilizzare un valore presente in un'altra cella, utilizza il 

carattere “e commerciale” (&) per concatenare, in questo modo: 

=CONTA.SE(intervallo;"<>"&A1) 

Se il valore della cella A1 è "100", dopo la concatenazione il criterio sarà "<>100" 

e CONTA.SE conterà le celle diverse da 100. 

12. Contare le celle diverse da tante cose 

 

 

Per contare le celle diverse da molte cose (vale a dire non uguali a x, y, z, ecc.) è 

possibile utilizzare una formula basata su CONTA.SE, 

MATR.SOMMA.PRODOTTO e CONTA.VALORI. 

Nell'esempio mostrato, la formula in G3 è: 

=CONTA.VALORI(B3:C12)-

MATR.SOMMA.PRODOTTO(CONTA.SE(B3:C12;esclusioni)) 

 

Normalmente, se hai solo un paio di cose che non vuoi contare, puoi usare 

CONTA.PIÙ.SE in questo modo: 

=CONTA.PIÙ.SE(intervallo;"<>blu";intervallo;"<>azz

urro") 

 

Ma questo potrebbe non essere molto comodo se hai un elenco di molte 

esclusioni, perché dovrai aggiungere una coppia intervallo/criterio per ogni cosa 

che non vuoi contare. Sarebbe molto più facile creare un elenco e passare un 

riferimento a questo elenco come criterio. Questo è esattamente quello che la 

formula di questa pagina fa. 

Questa formula utilizza l'intervallo denominato "esclusioni" (E3:E5) per riferirsi ai 

valori che non vogliamo conteggiare. 

Iniziamo contando tutti i valori nell'intervallo con CONTA.VALORI: 

=CONTA.VALORI(B3:C12) 

 

Successivamente, creiamo un conteggio di tutte le cose che non vogliamo contare 

con CONTA.SE, in questo modo: 

CONTA.SE(B3:C12;esclusioni) 

 

Poiché l’intervallo denominato "esclusioni" contiene più valori, CONTA.SE 

restituirà più risultati. Nell'esempio, riceviamo una matrice di valori come questa: 

{2.2.2} 

 

e utilizziamo MATR.SOMMA.PRODOTTO per sommare tutti gli elementi della 

matrice, ottenendo 6. Questo numero viene quindi sottratto dal totale originale per 

ottenere il risultato finale. 

L'utilizzo di MATR.SOMMA.PRODOTTO invece di SOMMA evita la necessità 

di utilizzare la sintassi delle formule matriciali. 

13. Contare le celle diverse da x e da y 

 
 

Per contare le celle diverse da questo e da quello, è possibile utilizzare la funzione 

CONTA.PIÙ.SE con più criteri. 

Nell'esempio mostrato, esiste un semplice elenco di tipi di frutta nella colonna B. 

Ci sono 10 celle totali, ognuna con un frutto e pochi sono duplicati. 

Per contare il numero di celle diverse da “mele” e da “pere”, la formula in E3 è: 

=CONTA.PIÙ.SE(frutta;"<>mele";frutta;"<>pere") 

 

In questo esempio "frutta" è un intervallo denominato che corrisponde a B3:B12. 

La funzione CONTA.PIÙ.SE conta le celle che soddisfano una o più condizioni. 

Tutte le condizioni devono essere soddisfatte per fare in modo che una cella venga 

contata. 

La chiave in questo caso è quella di utilizzare l'operatore "diverso da", che è <>. 

Per aggiungere altri criteri, basta aggiungere altre coppie di argomenti 

intervallo/criterio. 

Alternativa con MATR.SOMMA.PRODOTTO 

Anche la funzione MATR.SOMMA.PRODOTTO può contare le celle che 

soddisfano più condizioni. 

Per l'esempio precedente, la sintassi con MATR.SOMMA.PRODOTTO sarebbe 

stata: 

=MATR.SOMMA.PRODOTTO(frutta<>"mele")*(frutta<>"per

e")) 

14. Contare le celle vuote 

 

 

Per contare il numero di celle vuote, è possibile utilizzare la funzione 

CONTA.VUOTE. 

Nell'esempio, la cella attiva contiene questa formula: 

=CONTA.VUOTE(frutta) 

 

Dove per frutta si intende l’intervallo denominato per B3:B12. 

CONTA.VUOTE conta il numero di celle nell'intervallo che non contengono 

alcun valore e restituisce questo numero come risultato. Le celle che contengono 

testo, numeri, date, errori, ecc. non vengono conteggiate. 

È anche possibile contare facilmente le celle non vuote utilizzando la funzione 

CONTA.VALORI con lo stesso intervallo: 

=CONTA.VALORI(frutta) 

15. Contare le celle non vuote 

 

 

Per contare il numero di celle che non sono vuote, utilizza la funzione 

CONTA.VALORI. 

Nell'esempio, la cella attiva contiene questa formula: 

=CONTA.VALORI(frutta) 

 

Dove per frutta si intende l’intervallo denominato B3:B12. 

CONTA.VALORI è completamente automatica. Quando viene fornito un 

intervallo di celle, conta le celle che contengono numeri, testo, valori logici e errori. 

CONTA.VALORI non conta le celle vuote. 

Per contare le celle vuote, è possibile utilizzare la funzione CONTA.VUOTE 

come segue: 

=CONTA.VUOTE(frutta) 

Contare le celle con almeno un carattere 

Un problema con CONTA.VALORI è che conta anche le stringhe nulle restituite 

da formule (""). Se incorri in questo problema, puoi provare una formula come 

questa: 

=MATR.SOMMA.PRODOTTO(--LUNGHEZZA(intervallo)>0)) 

 

La funzione LUNGHEZZA restituisce un conteggio dei caratteri, in modo che 

MATR.SOMMA.PRODOTTO possa contare solo le celle contenenti almeno 1 

carattere. 

16. Contare le celle che iniziano con 

 

 

Per contare il numero di celle che iniziano con un determinato testo, è possibile 

utilizzare la funzione CONTA.SE. 

Nell'esempio, la cella attiva contiene questa formula: 

=CONTA.SE(B3:B12;"tab*") 

 

CONTA.SE conta il numero di celle dell'intervallo che inizia con il testo 

specificato confrontando il contenuto di ciascuna cella con lo schema "testo*", che 

viene fornito come criterio. Il simbolo "*" (l'asterisco) è un carattere jolly, che in 

Excel significa "qualsiasi numero di caratteri". Il conteggio delle celle che 

corrispondono a questo schema viene restituito come numero. 

 



  

 
Excel Academy 42 

17. Contare le celle che contengono x oppure 

y 

 

 

Per contare le celle che contengono un valore oppure un altro, si può utilizzare 

una colonna di supporto, oppure raggruppare il conteggio in una più complessa 

singola formula. 

Quando conti le celle con criteri "O", devi prestare attenzione a non raddoppiare il 

conteggio. Ad esempio, se stai contando le celle che contengono "cane" o "giallo", 

non puoi semplicemente sommare due funzioni CONTA.SE perché potresti 

raddoppiare le celle contenenti sia "cane" che "giallo". 

 

Soluzione a singola cella 

Per la soluzione a singola cella, è possibile utilizzare 

MATR.SOMMA.PRODOTTO con le funzioni VAL.NUMERO e TROVA. La 

formula nella cella F3 è: 

=MATR.SOMMA.PRODOTTO(--

((VAL.NUMERO(TROVA("cane";B3:B9)) + 

VAL.NUMERO(TROVA("giallo";B3:B9)))>0)) 

Questa formula si basa sulla formula vista nel capitolo CONTARE LE CELLE 

CONTENENTI DEL TESTO SPECIFICO che individua il testo all'interno di una cella: 

VAL.NUMERO(TROVA("cane"; B3:B9)) 

 

Quando viene fornito un intervallo di celle, questa porzione di formula restituirà 

una matrice di valori VERO/FALSO, un valore per ciascuna cella dell'intervallo. 

Poiché la usiamo due volte (una volta per "cane" e una volta per "giallo"), 

otteniamo due matrici. 

Quindi, sommiamo insieme queste matrici (con +), creando una nuova singola 

matrice di numeri. Ogni numero in questa matrice è il risultato della somma dei 

valori VERO e FALSO delle due matrici originali. Nell'esempio mostrato, la 

matrice apparirebbe così: 

{2.0.2.1.0.2.1} 

 

Dobbiamo sommare questi numeri, ma non vogliamo raddoppiare. Quindi 

dobbiamo assicurarci che un valore superiore a zero sia contato una sola volta. Per 

fare ciò, trasformiamo tutti i valori in VERO o FALSO con "> 0", quindi li 

tramutiamo in 1/0 con il doppio unario (--). 

Infine, MATR.SOMMA.PRODOTTO somma questi numeri. 

 

Soluzione con colonna di supporto 

Con una colonna di supporto per controllare ciascuna cella singolarmente, il 

problema è meno complesso. Possiamo usare CONTA.SE con due valori (forniti 

come "costante di matrice"). La formula in C3 è: 

=--(SOMMA(CONTA.SE(B3;{"*cane*"."*giallo*"}))>0) 

 

CONTA.SE restituirà una matrice che contiene due elementi: un conteggio per 

"cane" e un conteggio per "giallo". Per evitare il doppio conteggio, sommiamo gli 

elementi e forziamo il risultato a VERO/FALSO con ">0". Infine, convertiamo i 

valori VERO/FALSO in 1 e 0 con un doppio unario (--). 



  

 
Excel Academy 44 

Il risultato finale è 1 o 0 per ogni cella. Per ottenere un totale per tutte le celle 

dell'intervallo, sommerai semplicemente la colonna di supporto. 

18. Contare le celle che contengono errori 

 

 

Per contare il numero di celle che contengono errori, è possibile utilizzare la 

funzione VAL.ERR, racchiusa nella funzione MATR.SOMMA.PRODOTTO. 

Nell'esempio, la cella attiva contiene questa formula: 

=MATR.SOMMA.PRODOTTO(--VAL.ERR(B3:B8)) 

 

MATR.SOMMA.PRODOTTO accetta una o più matrici e calcola la somma dei 

prodotti dei numeri corrispondenti. Se viene fornita solo una matrice, la funzione 

somma solo gli elementi della matrice. 

La funzione VAL.ERR viene valutata per ogni cella dell’intervallo. Il risultato è una 

matrice di valori VERO/FALSO: 

{VERO.FALSO.FALSO.VERO.FALSO.FALSO} 

 

L'operatore “--“ (chiamato doppio unario) converte i valori VERO/FALSO in 1 e 

0. La matrice risultante è simile a questa: 

{1.0.0.1.0.0} 

MATR.SOMMA.PRODOTTO quindi somma gli elementi di questa matrice e 

restituisce il totale, che nell'esempio è 2. 

Nota: VAL.ERR conta tutti gli errori eccetto #N/D. Se vuoi contare anche gli 

#N/D, utilizza la funzione VAL.ERRORE invece di VAL.ERR. 

 

È anche possibile utilizzare la funzione SOMMA per contare gli errori. La struttura 

della formula è la stessa, ma deve essere inserita come formula matriciale 

(premendo Ctrl + Maiusc + Invio invece di Invio). La formula sarà simile a questa: 

{=SOMMA(--VAL.ERR(B3:B8))} 

19. Contare le celle che non contengono errori 

 

 

Per contare il numero di celle che non contengono errori, è possibile utilizzare le 

funzioni VAL.ERR e NON, racchiuse nella funzione 

MATR.SOMMA.PRODOTTO. 

Nell'esempio, la cella attiva contiene questa formula: 

=MATR.SOMMA.PRODOTTO(--NON(VAL.ERR(B3:B8))) 

 

MATR.SOMMA.PRODOTTO accetta una o più matrici e calcola la somma dei 

prodotti dei numeri corrispondenti. Se viene fornita una sola matrice, la funzione 

somma solo gli elementi della matrice. 

La funzione VAL.ERR viene valutata per ogni cella nell’intervallo. Senza la 

funzione NON, il risultato è una matrice di valori uguali a VERO o FALSO: 

{VERO.FALSO.FALSO.VERO.FALSO.FALSO} 

 

Con la funzione NON, il risultato viene invertito: 

{FALSO.VERO.VERO.FALSO.VERO.VERO} 

Questo corrisponde alle celle che non contengono errori nell’intervallo. 

L'operatore “--“ (chiamato doppio unario) converte i valori VERO/FALSO in 1 e 

0. La matrice risultante è simile a questa: 

{0.1.1.0.1.1} 

 

MATR.SOMMA.PRODOTTO quindi somma gli elementi di questa matrice e 

restituisce il totale, che nell'esempio è 4. 

È anche possibile utilizzare la funzione SOMMA per contare le celle senza errori. 

La struttura della formula è la stessa, ma deve essere inserita come una formula 

matriciale (premendo Ctrl + Maiusc + Invio invece di Invio). La formula sarà 

simile a questa: 

{=SOMMA(--NON(VAL.ERR(B3:B8)))} 

 

Non inserire le parentesi graffe {}: vengono immesse in automatico alla pressione 

di Ctrl + Maiusc + Invio. 

20. Contare le celle che contengono quattro 

caratteri 

 

 

Per contare il numero di celle che contengono un certo numero di caratteri, è 

possibile utilizzare la funzione CONTA.SE. 

Nell'esempio, la cella attiva contiene questa formula: 

=CONTA.SE(B3:B8;"????") 

 

CONTA.SE conta il numero di celle dell'intervallo che contengono quattro 

caratteri confrontando il contenuto di ciascuna cella con lo schema "????", che 

viene fornito come criterio per CONTA.SE. Il simbolo "?" è un carattere jolly che 

in Excel significa "qualsiasi carattere singolo", pertanto questo schema conta solo 

le celle che contengono quattro caratteri. Il conteggio delle celle che 

corrispondono a questo schema viene restituito come un numero che nell'esempio 

è 3. 

21. Contare le celle che contengono numeri 

negativi 

 

 

Per contare il numero di celle che contengono numeri negativi in un intervallo di 

celle, è possibile utilizzare la funzione CONTA.SE. 

Nell'esempio, la cella attiva contiene questa formula: 

=CONTA.SE(B3:B8;"<0") 

 

CONTA.SE conta il numero di celle di un intervallo che corrispondono al criterio 

fornito. In questo caso, il criterio è "<0", che viene valutato come "valori inferiori 

a zero". Il conteggio totale di tutte le celle nell'intervallo che rispondono a questo 

criterio viene restituito dalla funzione. 

È possibile aggiustare facilmente questa formula per contare le celle in base ad altri 

criteri. Ad esempio, per contare tutte le celle con un valore inferiore a -10, utilizza 

questa formula: 

=CONTA.SE(B3:B8;"<-10") 

 

Se vuoi utilizzare un valore presente in un'altra cella come criterio, utilizza il 

carattere “e commerciale” (&) per concatenare così: 

=CONTA.SE(intervallo;"<"&A1) 

 

Se il valore nella cella A1 fosse "-5", dopo la concatenazione il criterio sarà "<-5". 

22. Contare le celle che contengono numeri 

 

 

Per contare il numero di celle contenenti un numero, è possibile utilizzare la 

funzione CONTA.NUMERI. 

Nell'esempio, la cella attiva contiene questa formula: 

=CONTA.NUMERI(B3:B8) 

 

La funzione CONTA.NUMERI è completamente automatica. Conta il numero di 

celle nell'intervallo che contengono numeri e restituisce il risultato. 

 

Contare le celle che non contengono numeri 

Per contare il numero di celle di un intervallo che non contengono numeri, utilizza 

questa formula con MATR.SOMMA.PRODOTTO: 

=MATR.SOMMA.PRODOTTO(--NON(VAL.NUMERO(B3:B8))) 

23. Contare le celle che contengono numeri 

dispari 

 

 

Per contare le celle che contengono solo numeri dispari, è possibile utilizzare una 

formula basata sulla funzione MATR.SOMMA.PRODOTTO insieme alla 

funzione RESTO. 

Nell'esempio, la formula nella cella E3 è: 

=MATR.SOMMA.PRODOTTO(--(RESTO(intervallo;2)=1)) 

 

Questa formula restituisce 3 in quanto ci sono 3 numeri dispari nell'intervallo 

B3:B9 (che viene chiamato “intervallo” nella formula). 

La funzione MATR.SOMMA.PRODOTTO lavora in maniera nativa con le 

matrici. 

Una cosa che puoi fare facilmente con MATR.SOMMA.PRODOTTO è eseguire 

un test su una matrice utilizzando uno o più criteri, e quindi conteggiare i risultati. 

In questo caso, eseguiamo un test per trovare numeri dispari, utilizzando la 

funzione RESTO: 

RESTO(intervallo;2)=1 

RESTO restituisce il resto della divisione tra due numeri. In questo caso, il divisore 

è 2, quindi RESTO restituirà un resto di 1 per ogni numero intero dispari e un 

resto di zero per i numeri pari. 

All'interno di MATR.SOMMA.PRODOTTO, questo test viene eseguito su ogni 

cella di B3:B9, il risultato è una matrice di valori VERO/FALSO: 

{FALSO.FALSO.VERO.FALSO.VERO.FALSO.VERO} 

 

Dopo aver convertito i valori VERO/FALSO in numeri utilizzando il doppio 

unario, otteniamo: 

{0.0.1.0.1.0.1} 

 

MATR.SOMMA.PRODOTTO poi semplicemente somma questi numeri e 

restituisce 3. 

 


